unittest
{
	assert( [ "3", "2 1 2" ].parse.expand.solve == "3\n1\n0" );
	assert( [ "5", "1 2 3 4 5" ].parse.expand.solve == "5\n2\n1\n1\n1" );
	assert( [ "4", "1 3 3 3" ].parse.expand.solve == "4\n1\n0\n0" );
}



import std.algorithm;
import std.conv;
import std.range;
import std.stdio;
import std.typecons;

void main()
{
	stdin.byLineCopy.parse.expand.solve.writeln;
}

auto parse( Range )( Range input )
if( isInputRange!Range && is( ElementType!Range == string ) )
{
	input.popFront;
	auto as = input.front.split.to!( long[] );
	return tuple( as );
}

auto solve( long[] as )
{
	// 同じ数字のカードの枚数をカウント
	auto ns = as.sort().group.map!( ( x ) => x[ 1 ].to!( long ) ).array;
	
	// 最大操作回数を二分探索
	auto anss = new long[ as.length ];
	foreach( i, ref l; anss )
	{
		auto k = i.signed + 1;
		auto r = as.length.signed / k;
		while( l < r )
		{
			auto c = ( l + r + 1 ) / 2;
			if( solve2( ns, k, c ) ) l = c;
			else                     r = c - 1;
		}
	}
	return anss.to!( string[] ).join( "\n" );
}

auto solve2( long[] ns, long k, long m )
{
	// k個のバケツにm枚ずつ入れると考える
	auto c = 0L;
	auto t = 0L;
	foreach( n; ns )
	{
		auto tt = t + min( n, m );		// 一つのバケツでも複数のバケツでも同じ数字のカードを入れるのは合計m枚が上限
		if( m <= tt )
		{
			c++;
			if( k <= c ) return true;
		}
		t = ( tt ) % m;		// あふれた分は次のバケツに入れる
	}
	return false;
}
